# $Id: PKGBUILD 179542 2013-03-06 13:51:08Z bisson $
# Contributor: Wieland Hoffmann <the_mineo@web.de>
# Maintainer: Gaetan Bisson <bisson@archlinux.org>

pkgname=chromaprint
pkgver=0.7
pkgrel=4
pkgdesc='Library that implements a custom algorithm for extracting fingerprints from any audio source'
url='http://acoustid.org/chromaprint/'
arch=('i686' 'x86_64')
license=('LGPL')
depends=('ffmpeg')
makedepends=('cmake')
source=("https://github.com/downloads/lalinsky/${pkgname}/${pkgname}-${pkgver}.tar.gz"
        'ffmpeg.patch')
sha1sums=('6a961585e82d26d357eb792216becc0864ddcdb2'
          '4c99e8be678ccd0373ab1c0b52c068d6fabe21a2')

build() {
	cd "${srcdir}/${pkgname}-${pkgver}"

	patch -p1 -i ../ffmpeg.patch

	cmake \
		-DCMAKE_INSTALL_PREFIX=/usr \
		-DCMAKE_BUILD_TYPE=Release \
		-DBUILD_EXAMPLES=ON \
		.
	make
}

package() {
	cd "${srcdir}/${pkgname}-${pkgver}"
	make DESTDIR="${pkgdir}" install
}
